

Calibration of Models with Multi-Year Periods

Thomas F. Rutherford

Department of Economics, University of Colorado

rutherford@colorado.edu

June, 2001

1 Notation

Let:

n denote the number of years per period,

\bar{r} denote the *annual* interest rate,

r denote the *per period* interest rate, $r = (1 + \bar{r})^n - 1$,

$\bar{\delta}, \delta$ denote the *annual* and *per period* capital depreciation rates $\delta = (1 - \bar{\delta})^n - 1$,

\bar{g}, g denote the *annual* and *per period* steady-state growth rates,

R denote the base year capital earnings,

I denote base year investment.

2 GAMS/MPSGE Logic

The model we will calibrate involves investments in period t maturing in both periods t and $t + 1$. We assume that the base year investment of I produces A units of capital for use in the first period (during the first n years) and

B units of capital in the second period. The MPSGE production function for investment is:

$$\begin{array}{ll} \text{\$prod:i(t)} & \\ \text{o:pk(t)} & \text{q:A} \\ \text{o:pk(t+1)} & \text{q:B} \\ \text{i:p(t)} & \text{q:I} \end{array}$$

and the production function for capital services is:

$$\begin{array}{ll} \text{\$prod:k(t)} & \\ \text{o:pk(t+1)} & \text{q:((k0+A)*(1-delta))} \\ \text{o:rk(t)} & \text{q:(k0+A)} \\ \text{i:pk(t)} & \text{q:(k0+A)} \end{array}$$

where the initial endowment of capital is k_0 .

3 Calibration to a Steady-State Growth Path

The steady-state conditions for the model are:

1. Zero profit in investment – the value of inputs to investment (I) equals the value of capital produce both within and in the subsequent period:

$$p_K A + \frac{p_K B}{1+r} = I \quad (1)$$

2. Zero profit in capital supply – the value of a unit of capital at the start of the period equals the value of capital services provided plus the depreciated value of capital remaining at the start of the subsequent period:

$$p_K = r_K + \frac{1-\delta}{1+r} p_K \quad (2)$$

3. Steady-state capital stock – the capital at the start of the subsequent period equals the current capital stock incremented by the steady-state growth rate:

$$(K+A)(1-\delta) + B = K(1+g) \quad (3)$$

4. The return to capital and the capital stock are related through the rental price:

$$r_K(K + A) = R \quad (4)$$

We will typically assume that I and R are observed, while K is calibrated.

We assume that the annual data is consistent with a steady-state growth path, hence:

$$I = R \frac{\bar{g} + \bar{\delta}}{\bar{r} + \bar{\delta}}$$

We will calibrate the model so that the rental price on capital in the n -year model is identical to the return to capital in the annual model, i.e.

$$r_k = \bar{r} + \bar{\delta}$$

This assumption in turn defines the capital price by solving for p_K in 2:

$$p_K = (1 + r) \frac{\bar{r} + \bar{\delta}}{r + \delta}$$

We then have two equations in two unknowns which determine A and B :

$$\left(\frac{R}{r_k} - A \right) (g + \delta) = A(1 - \delta) + B \quad (5)$$

which reduces to

$$R \frac{g + \delta}{\bar{r} + \bar{\delta}} = A(1 + g) + B$$

and

$$I = p_K A + \frac{p_K}{1 + r} B \quad (6)$$

which reduces to:

$$I \frac{r + \delta}{\bar{r} + \bar{\delta}} = A(1 + r) + B$$

Solving two equations in two unknowns (A and B), we obtain:

$$A = \frac{I}{r - g} \left(\frac{r + \delta}{\bar{r} + \bar{\delta}} - \frac{g + \delta}{\bar{g} + \bar{\delta}} \right)$$

and

$$B = \frac{I}{r - g} \left[\frac{g + \delta}{\bar{g} + \bar{\delta}} (1 + r) - \frac{r + \delta}{\bar{r} + \bar{\delta}} (1 + g) \right]$$

4 An Illustrative Model

4.1 DOS Batch Program RUN.BAT

```
if exist simple.sol del simple.sol
echo    $setglobal n 1 >ndef
gams simple
echo    $setglobal n 2 >ndef
gams simple
echo    $setglobal n 4 >ndef
gams simple
echo    $setglobal n 8 >ndef
gams simple
echo    $setglobal n 16 >ndef
gams simple

gams compare
```

4.2 GAMS Program SIMPLE.GMS

```
$title  A simple model with multiyear time periods

$if exist ndef $include ndef
$if not setglobal n $setglobal n 3

scalar      n      Years per period           /%n%/,
            r0     Base year annual interest rate /0.05/,
            delta0 Annual depreciation rate     /0.07/,
            g0     Baseline annual growth rate  /0.02/,
            kvs    Capital value share in production /0.35/,
            shock  Shock to the capital stock /1/,
            r      Per period interest rate
            delta  Per period depreciation rate
            g      Per period growth rate;

r = (1+r0)**n - 1;
```

```

delta = 1 - (1-delta0)**n;
g = (1+g0)**n - 1;

set      yr      Years          /2000*2050/,
        t(yr)   Time periods in the model,
        t0(yr)  Initial year,
        tt(yr)  Terminal year;

parameter d(yr); d(yr) = ord(yr);

t0(yr) = yes$(ord(yr) eq 1);
t(t0) = yes;
loop(yr$t(yr), t(yr+n) = yes; if (ord(yr)+n > card(yr), tt(yr) = yes;));
display t,tt;

parameter      pref(yr)      Reference price index
              qref(yr)      Reference quantity index;

pref(yr) = 1/(1+r0)**(ord(yr)-1);
qref(yr) = (1+g0)**(ord(yr)-1);

scalar      a      Own-period investment
           b      Next-period investment
           i0      Base year investment
           kvs     Base year capital value share
           c0      Base year consumption
           pk0     Capital price index
           k0      Initial capital stock;

pk0 = (1+r) * (r0+delta0)/(r+delta);
i0 = kvs * (delta0+g0)/(delta0+r0);
c0 = 1 - i0;

*      Compute own- and next-period maturation shares:

a = i0/(r-g) * ( (r+delta)/(r0+delta0) - (g+delta)/(g0+delta0) );
b = i0/(r-g) * ( (g+delta)/(g0+delta0)*(1+r) - (r+delta)/(r0+delta0)*(1+g) );

```

```

k0 = kvs / (delta0 + r0) - a;
display r, g, delta;
display a, b, pk0;

parameter profit, kmkt;
profit = i0 - pk0 * a - pk0 * b / (1 + r);
kmkt = (kvs/(r0+delta0) - a) * (g + delta) - a * (1-delta) - b;
display profit, kmkt;

$ontext

$model:ramsey

$sectors:
    y(yr)$t(yr)      ! Output
    i(yr)$t(yr)      ! Investment
    k(yr)$t(yr)      ! Capital stock

$commodities:
    p(yr)$t(yr)      ! Output price
    pl(yr)$t(yr)     ! Wage rate
    rk(yr)$t(yr)     ! Return to capital
    pk(yr)$t(yr)     ! Capital price
    pkt              ! Terminal capital stock

$consumers:
    frank            ! Representative agent

$auxiliary:
    kt              ! Terminal capital stock

$prod:y(t)  s:1
    o:p(t)  q:1
    i:rk(t) q:kvs          p:pref(t)
    i:pl(t) q:(1-kvs)     p:pref(t)

$prod:i(yr)$t(yr)
    o:pk(yr+n)          q:b

```

```

        o:pkt$tt(yr)    q:b
        o:pk(yr)       q:a
        i:p(yr)        q:i0

$prod:k(yr)$t(yr)
        o:pk(yr+n)     q:((k0+a)*(1-delta))
        o:pkt$tt(yr)  q:((k0+a)*(1-delta))
        o:rk(yr)      q:((k0+a)*(r0+delta0))
        i:pk(yr)      q:(k0+a)

$demand:frank  s:0.5
        d:p(t)        q:(c0*qref(t))  p:pref(t)
        e:pl(t)       q:((1-kvs)*qref(t))
        e:pk(t0)      q:(shock*k0)
        e:pkt         q:(-k0)          r:kt

$constraint:kt
        sum(tt(yr), i(yr) * y(yr-n) - i(yr-n) * y(yr)) =e= 0;

$offtext
$sysinclude mpsgeset ramsey

y.l(t) = qref(t);
i.l(t) = qref(t);
k.l(t) = qref(t);
p.l(t) = pref(t);
pl.l(t) = pref(t);
rk.l(t) = pref(t);
pk.l(t) = pref(t) * pk0;
pkt.l = sum(tt, pref(tt)) * pk0 / (1+r);
kt.l = sum(tt, qref(tt)) * (1+g);

ramsey.iterlim = 0;
$include ramsey.gen
solve ramsey using mcp;

*      Reduce the capital stock by 25% and compute the
*      transition path:

```

```

shock = 0.75;
ramsey.iterlim = 8000;
$include ramsey.gen
solve ramsey using mcp;

*      Compute a cubic spline approximation to produce annual
*      data based on the n-year data:

variables      x(yr), alpha, beta, gamma, phi, obj;

parameter      x0(yr)  Reference value;

equations      objdef  Defines square deviation
               cubic   Cubic spline fit;

objdef..       obj =e= sum(t, sqr(x(t)-x0(t)));

cubic(t)..     x(t) =e= alpha + beta * d(t) + gamma * d(t) * d(t)
               + phi * d(t) * d(t) * d(t);

model cubicfit / objdef, cubic /;

$setglobal prefix "'%n%.'"
file ksol /simple.sol/; ksol.ap = 1; put ksol;

$batinclude fit y
$batinclude fit i
$batinclude fit k
$batinclude fit p
$batinclude fit pl
$batinclude fit rk
$batinclude fit pk

```

4.3 GAMS Program FIT.GMS (called from SIMPLE.GMS)

```
x0(t) = %1.1(t);
```



```

solve cubicfit using nlp minimizing obj;

%1.1(yr) = alpha.1 + beta.1 * d(yr)
          + gamma.1 * d(yr) * d(yr)
          + phi.1 * d(yr) * d(yr) * d(yr);

$setglobal prefix "'%n%.%1.'"
$libinclude gams2txt %1.1

```

4.4 GAMS Program COMPARE.GMS

```

$title Compare Results

set      ti /1,2,4,8,16/;
parameter sol /
$include simple.sol
/;

parameter      capstock      Capital stock adjustment,
               invest        Investment adjustment;

set yr /2000*2050/;

scalar      g0      Baseline annual growth rate      /0.02/;

parameter      qref(yr)      Reference quantity index;

qref(yr) = (1+g0)**(ord(yr)-1);

capstock(yr,ti) = 100 * (sol(ti,"k",yr)/qref(yr)-1);
invest(yr,ti) = 100 * (sol(ti,"i",yr)/qref(yr)-1);


set t1(yr)/2000,2010,2020,2030,2040,2050/;

$setglobal domain yr
$setglobal labels t1

```

```
$setglobal gp_opt0 'set key outside'  
$libinclude plot capstock  
$libinclude plot invest
```

Capital stock adjustment



Investment adjustment

